# Maintainer: 

pkgname=perl-text-patch
pkgver=1.8
pkgrel=12
pkgdesc="Patches text with given patch"
arch=('any')
url="https://search.cpan.org/dist/Text-Patch"
license=('PerlArtistic' 'GPL')
depends=('perl')
checkdepends=('perl-text-diff')
options=('!emptydirs')
source=("https://search.cpan.org/CPAN/authors/id/C/CA/CADE/Text-Patch-$pkgver.tar.gz")
sha256sums=('eaf18e61ba6a3e143846a7cc66f08ce58a0c4fbda92acb31aede25cb3b5c3dcc')

build() {
  cd Text-Patch-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd Text-Patch-$pkgver
  make test
}

package() {
  cd Text-Patch-$pkgver
  make DESTDIR="$pkgdir" install
}
